/*
CpGears v1.0

Copyright (c) 2010 Yves Riel http://www.flash-factor.com

Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in
all copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
THE SOFTWARE.
*/ 
package cpgears.movie.container {
	import flash.utils.Dictionary;
	import cpgears.helpers.StringUtil;
	
	/**
	 * Enumeration class containing the Captivate Container Types
	 * 
	 * @author Whyves
	 */
	public class ContainerType{
		
		// TRICKY: The static dictionairy must be the first one declared.
		private static var mEnumeration:Dictionary = new Dictionary();
		
		/**
		 *  The widget's container is the Slide itself.
		 */
		public static const SLIDE:ContainerType = new ContainerType("Slide");
		
		/**
		 *  The widget's container is the Movie since the widget spans the rest of the project.
		 */
		public static const MOVIE:ContainerType = new ContainerType("Movie");
		
		/**
		 *  The widget's container is a slidelet.
		 */
		public static const SLIDELET:ContainerType = new ContainerType("Slidelet");

		/**
		 *  The widget's container is a Zoom Destination Slide.
		 */
		public static const ZOOM_DESTINATION_SLIDE:ContainerType = new ContainerType("Zoom Destination Slide");
		
		/**
		 *  The widget's container is a Zoom Destination Slidelet.
		 */
		public static const ZOOM_DESTINATION_SLIDELET:ContainerType = new ContainerType("Zoom Destination Slidelet");
		
		private var mName:String = null;
		
		/**
		 * Creates a container type.
		 * 
		 * @param	name The container internal name.
		 */
		public function ContainerType(name:String) {
			if (StringUtil.isNullOrEmpty(name)) throw new ArgumentError("'name' parameter cannot be null or empty.");
			mName = name;
			mEnumeration[name] = this;
		}

		/**
		 * Returns the container which internal name matches the parameter.
		 * 
		 * @param	containerName The internal name of the container.
		 * @return The container matching the internal name.
		 */
		public static function getContainerFromName(containerName:String):ContainerType {
			return (mEnumeration.hasOwnProperty(containerName)) ? mEnumeration[containerName] : new ContainerType(containerName);
		}
		
		/**
		 * Returns the container's internal name
		 */
		public function get value():String {
			return mName;
		}
	}
}